
การปรับสเกลหนาจอแบบพอดีกับกราฟกคาน 

ผศ.สรกานต  ศรตีองออน 

ปญหาตอมาในการแสดงกราฟกคาน  กรณีเรากําหนดสเกลในการแสดงผลที่หนาจอแบบ 
คงที่คือ  เม่ือคานนั้นมคีวามยาวรวมมากกวาสเกลที่ตั้งไว  ก็จะทําใหแสดงไดไมเต็มภาพ  วิธีแกมี  2 
วิธี คือ 

1.  ปรับสเกลแบบ Best Fit คือหนาจอใหพอดีกับภาพที่แสดงเสมอ 
2.  ใชวิธีเล่ือนจอภาพได 

บทความนีจ้ะแสดงวิธีที่  1  โดยกรณีของกราฟกคานความยาวจะแปรเปล่ียนเฉพาะแนวแกน  X 
เทานั้นกอน สวนแกน Y จะใชหลักการ 3/4 เทาของแกน X 

แนวคิด 
เราใชวิธีกําหนดสเกลหนาจอแบบยืดหยุน คือ 

q  ตรวจสอบความยาวของภาพที่จะแสดงกอน แลวปรับสเกลแกน X  ทีหลัง 
q  กํานดพกิัดขอบซายใหเริ่มที่  -2  หนวย  เพราะเริ่มพล็อตภาพท่ีพิกัด  0  นั่นคือเวนระยะหาง 

ออกไปทางซายจากภาพ 2 หนวย (ซึ่งหนวยทีใ่ชในที่นี้คือ เมตร) 
q  พิกัดขอบขวาจะเทากับความยาวรวมของคาน+2 หนวย นั่นคือเวนระยะหางออกไปทางขวา 

จากภาพ 2 หนวย 
q  สเกลแกน Y เทากับ 3/4 เทาของสเกลแกน X โดยพิกดัขอบลางเริ่มที่ -2 หนวย 
q  ดังนั้น พกิัดขอบบนเทากับ (3 x ความยาวสเกลแกน X \ 4) - 2 

การโปรแกรม 
1.  ปรับรหัสของโปรแกรมเมื่อบทความตอนที่แลวดังรายละเอียดดานลาง  โดยมี  Sub  เพ่ิมเติมซ่ึง 

เปนจุดสําคัญคือ FitScreen 

Option Explicit 
Dim Xmin As Single: Dim Xmax As Single 
Dim Ymin As Single: Dim Ymax As Single 
Dim StepX As Single: Dim StepY As Single 
Const N = 3 ' สมมุติจํานวนชวงคานเทากับ 3



Dim L(0 To N + 1) As Single ' ความยาวแตละชวงคาน 
Dim SumL As Single ' ผลรวมความยาวของแตละชวงคาน 
' รูปแบบคาน : 1 = ไมมีชวงยื่น, 2 = ปลายยื่นซาย 
' 3 = ปลายยื่นขวา, 4 = ปลายยื่นสองดาน 
Dim BeamType As Integer 
Dim S As Integer ' คาเริ่มตนของลูป 
Dim E As Integer ' คาจบของลูป 

Sub PlotBeam() 
Dim I As Integer 
Dim X As Single: Dim Y As Single 
Dim d As String 

' สรางกราฟกคานแตละชวง 
X = 0: Y = Ymax – 5 
For I = S To E 

Line (X, Y + 0.1)-(X + L(I), Y), QBColor(1), B 
X = X + L(I) 

Next I 
' เสนบอกความยาวแตละชวง 
X = 0: Y = Ymax – 7 
Line (X, Y)-(X + SumL, Y), QBColor(4) 
Line (X, Y + 0.5)-(X, Y - 0.5), QBColor(4) 
For I = S To E 

X = X + L(I) 
Line (X, Y + 0.5)-(X, Y - 0.5), QBColor(4) 

Next I 
' แสดงคาความยาวแตละชวง 
X = 0: Y = Ymax – 7 
For I = S To E 

d = Format(L(I), "0.00") 
CurrentX = X + (L(I) / 2) - (TextWidth(d) / 2)



CurrentY = Y - TextHeight(d) 
Print d 
X = X + L(I) 

Next I 
' เสนแรงปฏิกิริยา 
X = 0: Y = Ymax – 5 
Select Case BeamType 

Case 2, 4 ' ปลายยื่นซาย, ปลายยื่นสองดาน 
X = X + L(0) 

End Select 
DrawWidth = 3 
Line (X, Y)-(X, Y - 1), QBColor(1) 
Line (X, Y)-(X - 0.1, Y - 0.1), QBColor(1) 
Line (X, Y)-(X + 0.1, Y - 0.1), QBColor(1) 
DrawWidth = 1 
For I = 1 To N 

X = X + L(I) 
DrawWidth = 3 
Line (X, Y)-(X, Y - 1), QBColor(1) 
Line (X, Y)-(X - 0.1, Y - 0.1), QBColor(1) 
Line (X, Y)-(X + 0.1, Y - 0.1), QBColor(1) 
DrawWidth = 1 

Next I 
End Sub 

Private Sub Form_Activate() 
Dim Row As Integer: Dim Col As Integer 
Dim NumX As Single: Dim NumY As Single 
Dim X As Single: Dim Y As Single 

Me.Scale (Xmin, Ymax)-(Xmax, Ymin) ' กําหนดสเกลการแสดงผล 
NumX = (Abs(Xmax - Xmin) / StepX) + 1 ' หาจํานวนจดุกริดในแกน X



NumY = (Abs(Ymax - Ymin) / StepY) + 1 ' หาจํานวนจดุกริดในแกน Y 
Cls ' เคลียรหนาจอ 
' พล็อตจุดกรดิ 
X = Xmin: Y = Ymin 
For Row = 1 To NumX 

For Col = 1 To NumY 
PSet (X, Y), QBColor(4) 
Y = Y + StepY 

Next Col 
X = X + StepX: Y = Ymin 

Next Row 
' ลากเสนแกนอางอิงพิกัด (0,0) 
Line (0, 0)-(1.5, 0), QBColor(4): Line (0, 0)-(0, 1.5), QBColor(4) 
' เรียกโปรแกรมยอยสําหรับสรางกราฟกคาน 
PlotBeam 

End Sub 

Sub FitScreen() 
Dim I As Integer 

' หาคาเริ่มตนและคาจบของลูป 
Select Case BeamType 

Case 1 ' ไมมีชวงยื่น 
S = 1: E = N 

Case 2 ' ปลายยื่นซาย 
S = 0: E = N 

Case 3 ' ปลายยื่นขวา 
S = 1: E = N + 1 

Case 4 ' ปลายยื่นสองดาน 
S = 0: E = N + 1 

End Select 
' หาผลรวมความยาวของแตละชวงคาน



SumL = 0 
For I = S To E 

SumL = SumL + L(I) 
Next I 
' กําหนดสเกลหนาจอแกน X เซ็ตขอบซายที่ -2 m เสมอ 
Xmin = -2: Xmax = SumL + 2 
' กําหนดสเกลหนาจอแกน Y เทากับ 3/4 เทาของแกน X โดยเซ็ตขอบลางท่ี -2 m เสมอ 
Ymin = -2: Ymax = (3 * (Xmax + Abs(Xmin)) \ 4) - Abs(Ymin) 

End Sub 

Private Sub Form_Load() 
Dim I As Integer 

Me.WindowState = 2 ' ขยายฟอรมใหเต็มจอภาพ 
Me.ScaleMode = 0 ' กําหนดสเกลหนาจอเอง 
Me.BackColor = vbWhite ' สีฉากหลังเปนสีขาว 
Me.AutoRedraw = True ' จัดเก็บการแสดงผลในหนวยความจํา 

' ******************** ขอมูลที่สมมุติขึ้น ************************ 
' สมมุติคานปลายยื่นซาย 
BeamType = 2 
L(0) = 2: L(1) = 12: L(2) = 12: L(N) = 10 
' ************************************************************* 
FitScreen 
StepX = 1: StepY = 1 ' กําหนดระยะในการแสดงจุดกรดิ 
Call Form_Activate ' ไปท่ีเหตุการณ Activate 

End Sub 

Private  Sub  Form_MouseMove(Button  As  Integer,  Shift  As  Integer,  X  As  Single,  Y  As 
Single) 

' แสดงตําแหนงพิกัดที่ StatusBar ทศนิยมหนึ่งตําแหนง



stbMain.SimpleText  =  "พิกัดแกน  X  =  "  &  Format(X,  "0.0")  &  ",  พิกัดแกน  Y  =  "  & 
Format(Y, "0.0") 
End Sub 

2.  เม่ือรันโปรแกรม จะแสดงผลดังรูป (สังเกตวาผูเขียนสมมุติ BeamType = 2 หมายความวา คาน 
ปลายยื่นซาย และความยาวคานรวมเทากับ 36 m เพ่ือใหสเกลแกน X เทากับ |-2|+36+2 เทากับ 
40 m ดังนั้น สเกลแกน Y จึงเทากับ |-2|+26+2 เทากับ 30 m เพ่ือใหเห็นถึงอัตราสวน 3/4 แบบ 
ชัดเจน ) 

หมายเหตุ 
q  ลองปรับความยาวของคานเปนรูปแบบตางๆ  แลวดูผลการแสดงภาพนะครับ  จะเขาใจ 

ยิ่งขึ้น 
q  การแสดงผลแบบพอดีกับภาพนี้  กรณคีวามยาวมากๆ  ก็จะมีปญหา  ดังนั้น  โปรแกรมสวน 

ใหญจะมีทางเลือกในการแสดงผลอีกแบบหนึ่งคือ การเล่ือนจอภาพได  ซ่ึงเทคนคิการเขียน 
โปรแกรมจะซับซอนขึ้น



q  กรณกีารพล็อตที่ความยาวแปรเปล่ียนทั้งแกน X และแกน Y เชน กราฟกโครงขอหมุน จะ 
มีวิธีปรับสเกลอีกแบบหนึ่งที่ตางออกไปเล็กนอย 

q  สําหรับขอ 2 และ 3 ผูเขียนจะนําเสนอในตอนตอๆ ไป นะครับ 

